
The “Annotate” facility of SAS/GRAPH® has great potential, 
which is usually left unfulfilled due to lack of knowledge on the 
part of the programmer (or perhaps fright).  This paper 
introduces the basics of “Annotate”, with some simple 
examples.  

INTRODUCTION  
 
The standard SAS/Graph procedures – GPLOT, GCHART and 
the rest – do an excellent job of providing the most commonly 
required graphics facilities. If you need anything further adding 
to your plot, Annotate is the facility that enables you to do it.  
 
“Annotate” is a low-level facility. Working with Annotate is, 
perhaps, the SAS/Graph equivalent to programming in 
Assembler. It’s basically pretty simple, but you usually have to 
work quite hard to achieve anything.  
 
OTHER WAYS OF ENHANCING SAS GRAPHICS 
 
You should be aware that Annotate is not the only tool in the 
box. Others include: 
 

• ODS and Device Drivers (some of which can bring a 
degree of interactivity to your plots). See especially 
“Types of Web Presentations Available” in the SAS 
Online documentation 

• Macros can often be used to good effect e.g. to 
generate SYMBOL statements that depend on the 
data (so that, say, data for cities within the same 
country is plotted using symbols of the same colour) 

• PROC GREPLAY to generate montages of plots,  to 
rotate, flip and/or overlay plots, etc. 

• DSGI – the Data Step Graphics Interface – for 
generating graphics directly from a data step 

• PROC GFONT enables you to define your own fonts, 
which can be used e.g. as plotting symbols 

• There’s also a SAS Graphics Editor, which can be 
used for manually altering individual plots. 

 
THE BASIC MECHANISM 
 
The Annotate facility is available to most SAS/Graph 
procedures - GPLOT, GCHART, G3D, GSLIDE, etc. There’s 
also a GANNO procedure that displays the annotations and 
nothing else. 
 
The annotations are supplied in a separate data set, which has 
to be prepared in an earlier data step. It is then specified to the 
graphics procedure as an extra input thus: 
 
   proc gplot data=mydata  
           annotate=myfancystuff; 
           plot a*b=c; 
    run; 
 
THE ANNOTATE DATA SET 

The structure of the “Annotate data set” is defined by the SAS 
Institute and non-negotiable. The names of the variables are 
fixed. Some of the variables have SAS-defined sets of 
permissible values.  

The Annotate dataset can be data-dependent e.g. for the 
embellishment in some way of each individual point plotted. 

It is also possible to have fixed annotations, which could be 
stored in a permanent data set. Here’s an example of such 
fixed annotations, from a geochemical application: 

 

Since all Annotate data sets have the same structure, they can 
be concatenated without problems. So it would be possible to 
append the data set specifying the above annotations to a data 
set specifying data-dependent annotations – for example, to 
label each point with details of how and where the reading was 
taken. The resulting larger dataset could then be specified as 
the Annotate data set for a PROC GPLOT. 

EXAMPLE: A SCATTER PLOT TO ANNOTATE 
 
We begin by producing some annotations for this scatter plot: 

 
 



 
 

The code that produced this was simply: 

symbol1 font=marker value='R' h=2 color=red; 
symbol2 font=marker value='Q' h=2 color=blue; 
 
proc gplot data=demog; 
  plot weight*height=gender; 
run; 
 
ANNOTATING FROM FIRST PRINCIPLES 
 
There is more than one way of generating an Annotate data 
set. Here is the hard way: 
 
data anno_a; 
  length function style $8 text $12; 
  retain function 'label' xsys ysys '2'  
         hsys '3' style 'centxi' 
         size 2 when 'a' color 'green'  
         position '9'; 
  set demog(keep=height weight); 
  x=height; 
  y=weight; 
  text=left(put(weight,roman8.)); 
  drop height weight; 
run; 
 
This adorns each point plotted with text which expresses in 
Roman numerals the “weight” value being plotted. Note that: 
 
• We have to know the names of the Annotate variables we 

are using. 
• We have to explicitly “retain” most of them. 
• Each record written will have FUNCTION=’label’, which is 

the function for producing text strings. 
• Variables XSYS and YSYS both have the value ‘2’. This 

specifies our co-ordinate system as “data values” i.e. we 
specify any (x,y) location in the plot area by passing to 
“Annotate” the height and weight values corresponding to 
that location. 

• HSYS has the value ‘3’, meaning that heights are 
measured as percentages of the plot area. Since SIZE=2, 
our character height will be 2% of the plot area. The font 
used is CENTXI, coloured GREEN. 

• A POSITION of ‘9’ specifies that the text will be positioned 
slightly to the south-east of the data point to which it 
relates. 

• WHEN=’A’ means that the annotations will be plotted 
AFTER the standard graphics. 

 
Now we are ready to generate the annotated plot. 
 
proc gplot data=demog annotate=anno_a; 
  plot weight*height=gender; 
run; 
 
yields: 
 
 
 
 
 
 
 
 
 
 

 
 
ANNOTATING USING MACROS 
 
The process of generating the Annotate dataset can be made 
slightly easier by using some macros kindly provided by SAS 
Institute. Let’s annotate the same plot slightly differently, and 
this time using the macros. 
 
%annomac; 
data anno_b; 
  %dclanno; 
  length text $12;  
  set demog(keep=height weight); 
  %system(2,2,3);  
  text=left(put(weight,roman8.)); 
  %label(height,weight,text,steel, 30,180,2,swissb,F); 
  drop height weight; 
run; 
 
There are several changes here compared with our previous 
version: 
 
• We begin by invoking the %ANNOMAC macro. This 

makes all the other Annotate macros available for use. 
• Within the data step, we use the %DCLANNO macro to 

define ALL the Annotate variables. We don’t have to 
bother with any LENGTH or RETAIN statements, with the 
sole exception of the TEXT variable. We still have to set 
the length of this, because Annotate cannot know in 
advance how long our text strings are going to be. 

• The %SYSTEM macro is then used to set the values of 
XSYS, YSYS and HSYS. It is best not to rely on the 
default values for the co-ordinate systems!  

• The %LABEL macro is used to print our text strings, and it 
has a lot of parameters. Here we have specified a colour 
of STEEL, the text strings to be printed at an angle of 30° 
with each individual character rotated through 180°, and a 
height of 2. We have to specify all these parameters, but it 
is permissible to specify them as missing values and so 
accept the default e.g.  

 
         %label(height,weight,text,steel,.,.,2,swissb,F); 
 
However, if we leave the silly angles specified, the annotated 
plot generated using the ANNO_B dataset looks like this: 

THE EASIEST WAY OF ALL 
 
Label your points using the POINTLABEL option of the 
SYMBOL statement!  
 
Moral: Don’t be in too much of a hurry to use Annotate; there 
may sometimes be softer options. 

THE ANNOTATE DICTIONARY 
 
Once you have grasped the basics, using Annotate becomes 
more or less a stamp-collecting exercise. The key part of the 
documentation is the “Annotate Dictionary”. This contains 
several different kinds of entry, and it is important to be aware 
of what kind of entry you are looking at.  
 

• MACROS are what will mainly appear in your code. 
• FUNCTIONS are the corresponding values of the 

Annotate variable FUNCTION, which you don’t 
normally need to know. Most of the time there is a 
one-to-one correspondence between macros and 
functions – but not always. The documentation of 
each function lists which variables are relevant to it. 

• VARIABLES are the internal Annotate variable 
names, which you don’t normally need to know (but 
there are exceptions e.g. ZSYS). Note that the same 
variable is liable to be used for different purposes by 
different macros and functions e.g. the STYLE 
variable is used by %LABEL to specify a font, but by 
several other macros to specify a fill pattern. In such 
a situation, you cannot rely on the values you expect 
persisting as defaults, so it is usually best to specify 
everything explicitly. 

 
FIELD GUIDE TO THE ANNOTATE MACROS 
 
General macros 
 
• %ANNOMAC defines all the other Annotate macros. 
• %DCLANNO declares all the standard Annotate variables 

required in an annotate data set. This includes all the 
LENGTH and RETAIN statements you need, with the sole 
exception that you have to declare the length of the TEXT 
variable for yourself (so your text strings can be as long as 
you need them). 

• %COMMENT writes a comment to the Annotate data set. 
• There is a DEBUG function (though no corresponding 

macro), which you will obviously not need! 
 
Basic macros 

 
• %LABEL writes a text string to the graphics area. 
• %DRAW draws a line from the current position. 
• %MOVE alters the current position without drawing 

anything. 
• %LINE draws a line between any two points. 
• %POLY, %POLY2 and %POLYCONT are used for 

drawing polygons, which can be filled. 
• %BAR and %BAR2 draw rectangles, which can be filled. 
• %RECT draws a non-fillable rectangle. 
 
Specialised macros 
 
These are intended for use in annotating pie charts and maps. 
 
• %CIRCLE draws a circle, which can be filled. 
• %SLICE draws a sector of a circle, which can be filled. 
• %PIEXY moves to a position at the edge of a pie chart. 
• %CENTROID moves to the centroid of a map area. 
• %MAPLABEL specifies a dataset containing the labels 

with which a map is to be annotated. 
 
Macros related to co-ordinate systems 
 
• %SYSTEM defines the co-ordinate systems to be used for 

X, Y and H. For permissible values, see the 
documentation. You can speciify either absolute or relative 
co-ordinate systems, which can be based on the data 
area, the graphics output area, or the procedure output 
area. There are values that signify, for example, “use data 
values as co-ordinates”, or “percentages of graphics 
output area”. NB %SYSTEM does not set the co-ordinate 
system for the Z axis used in 3D plots – you have to set 
the value of the ZSYS variable directly. 

• %FRAME draws a frame around the area corresponding 
to the co-ordinate system chosen i.e. the data area, the 
graphics output area or the procedure output area. 

• Annotate “remembers” the most recent “text co-ordinates” 
used, and also the most recent “control co-ordinates” used 
(i.e. for anything other than text). Several macros are 
available for doing non-standard things with these 
remembered co-ordinates: %DRAW2TXT, %CNTL2TXT, 
%TXT2CNTL, %SWAP. You need to explore these 
advanced techniques if you want to underline text, for 
example. Note that you cannot access the remembered 
co-ordinates directly, since they do not correspond to any 
variables in the Annotate data set. 

 
Miscellaneous macros 
 
• %PUSH and %POP are used for manipulating the LIFO 

stack of co-ordinate pairs maintained by Annotate. You are 
unlikely to need to do this. 

• %SCALE and %SCALET are used for scaling co-
ordinates. They do not affect text functions. 

• %SEQUENCE is used to specify whether the Annotate 
graphics are generated BEFORE or AFTER the standard 
procedure graphics. All it does is to set the value of the 
WHEN variable. 

ANNOTATING A BAR CHART 

Here is a bar chart which we will annotate. 
 
axis1 label=none major=none minor=none style=0 value=none; 
axis2 label=none value=(font=swissb height=2); 
proc gchart data=profile; 
  vbar gender 
       / type=sum    
         sumvar=freq 
         subgroup=agerange 
         width=20 



 
 

         raxis=axis1 

         maxis=axis2; 
run; 

 
 
This chart shows the makeup of a company’s workforce, in 
terms of gender and age group. We will add to it the actual 
figures corresponding to each coloured area. The figures will be 
in red, and surrounded by purple circles. 
 
Annotating bar charts need not be as difficult as we are making 
it here; this example is intended to be educational rather than of 
practical value! 
 
Our code to generate the Annotate dataset looks like this: 

%annomac;  
data anno_c; 
  %dclanno; 
  when='after'; 
  set profile(keep=gender agerange freq); 
  %system(2,2,3); 
  midpoint=gender;  
  subgroup=agerange;   
  function='label';   
  text=left(put(freq,2.)); 
  style='swissb';         
  size=3;     
  color='red'; 
  position='5'; 
  output;   
  %circle(.,.,4,purple); 
  drop gender agerange freq; 
run; 
 
Points to note: 
 
• We are generating 2 records in the Annotate data set for 

each record in the input data set – one tor the “label” text, 
and one for the circle that surrounds it. We therefore need 
an explicit OUTPUT statement to write the first record. The 
second is written by %CIRCLE. 

• We specify the position for the text using an alternative co-
ordinate system which Annotate provides for use with bar 
charts – GROUP, MIDPOINT, SUBGROUP. This system 
cannot be used with the standard Annotate macros, so we 
cannot use %LABEL here. (Note also that this co-ordinate 
system cannot be specified using %SYSTEM, but it is 
compatible with system ‘2’ – ‘data values’ – as specified 
here.)  

• We do, however, manage to use the %CIRCLE macro, by 
specifying missing values for the X and Y co-ordinates, so 
that the circle is plotted in the default position. 

• We are obliged to specify variables FUNCTION, STYLE, 
SIZE and COLOR afresh for each text label (rather than 
initialising them once and for all, as with our scatter plot 
example earlier). The reason is that each of these 
variables is also used by the %CIRCLE macro, though not 
necessarily for the same purpose – SIZE is the radius of 
the circle, and STYLE the fill pattern. 

• We have to explicitly set the WHEN variable, otherwise 
most of our annotations would be hidden underneath the 
bars. 

 
We can now generate the annotated bar chart thus: 
 
axis1 label=none major=none minor=none style=0 value=none; 
axis2 label=none value=(font=swissb height=2); 
proc gchart data=profile; 
  vbar gender 
       / type=sum    
         sumvar=freq 
         subgroup=agerange 
         width=20 
         annotate=anno_c 
         raxis=axis1 
         maxis=axis2; 
run; 
 

 



 
 

ANNOTATION OUTSIDE THE PLOT AREA CONTACT INFORMATION  
Your comments and questions are valued and encouraged.  
Contact the author at: Here’s an example of another application of annotation, again 

from the world of geochemistry:  
Author Name Bob Newman 

 Company: Amadeus Software Limited 

 

Address: Mulberry House,  
9 Church Green, 
Witney, 
Oxon OX28 4AZ 

 Work Phone:  +44 (0) 1993 848010 
 Email:  bob.newman@amadeus.co.uk 
  Web:  www.amadeus.co.uk 
 
SAS and all other SAS Institute Inc. product or service names are 
registered trademarks or trademarks of SAS Institute Inc. in the USA 
and other countries. ® indicates USA registration.   
 
Other brand and product names are trademarks of their 
respective companies.  
 

Here the axes have been heavily annotated, dividing them into 
ranges marked high/medium/low etc. The positions of these 
annotations were specified as percentages of the data area, but 
some of the percentages required were over 100%, which 
Annotate regards as invalid. Annotate will only allow 20 errors 
before it refuses to process any more data. The above 
annotations would generate considerably more than 20 such 
errors. So how was this plot produced? 
 
The answer is that there is an undocumented technique for 
overriding that error limit of 20. Include the following near the 
beginning of the datastep that generates your Annotate data 
set: 
 

function=’seterror’; 
size=-1; 
output; 
 

After this, Annotate will keep going however ill-conceived it 
feels your annotations may be. 

SUMMARY 
 
Annotate is not as fierce as its reputation. It is a useful 
technique to add to your SAS/Graph repertoire, since it allows 
you to do many things that cannot be done any other way. This 
paper has presented the basics of Annotate, together with 
several simple examples, but the possibilities are limited only 
by your imagination. 

